NMST 434, Exercise session VI: M-estimators

April 1, 2019

```
SetOptions[Plot, BaseStyle → FontSize → 16];
```

Example: S_n as an M-estimator

```
Clear[M];
$Assumptions = τ > 0 && σ2 > 0;
ψ[μ_, σ2_, τ_] = {x - μ, (x - μ)^2 - σ2, √σ2 - τ};
D[ψ[μ, σ2, τ], {{μ, σ2, τ}}] /. x → μ
Ai = Inverse[%] /. σ → √σ2;
B =
  ((Outer[Times, ψ[μ, σ2, τ], ψ[μ, σ2, τ]] // Expand) /. Table[x^(4-j) → M[4 - j], {j, 0, 4}])
Ai.B.Transpose[Ai] /. {M[1] → μ, M[2] → σ2 + μ^2, σ2 → τ^2} // Simplify // MatrixForm
```

$$\left\{\{-1, 0, 0\}, \{0, -1, 0\}, \left\{0, \frac{1}{2\sqrt{\sigma2}}, -1\right\}\right\}$$

$$\Big\{\Big\{\mu^2 - 2\,\mu\,M[1] + M[2],\ -\mu^3 + \mu\,\sigma2 + 3\,\mu^2\,M[1] - \sigma2\,M[1] - 3\,\mu\,M[2] + M[3],$$
$$-\mu\sqrt{\sigma2} + \mu\,\tau + \sqrt{\sigma2}\,M[1] - \tau\,M[1]\Big\},\ \Big\{-\mu^3 + \mu\,\sigma2 + 3\,\mu^2\,M[1] - \sigma2\,M[1] - 3\,\mu\,M[2] + M[3],$$
$$\mu^4 - 2\,\mu^2\,\sigma2 + \sigma2^2 - 4\,\mu^3\,M[1] + 4\,\mu\,\sigma2\,M[1] + 6\,\mu^2\,M[2] - 2\,\sigma2\,M[2] - 4\,\mu\,M[3] + M[4],$$
$$\mu^2\sqrt{\sigma2} - \sigma2^{3/2} - \mu^2\,\tau + \sigma2\,\tau - 2\,\mu\sqrt{\sigma2}\,M[1] + 2\,\mu\,\tau\,M[1] + \sqrt{\sigma2}\,M[2] - \tau\,M[2]\Big\},$$
$$\Big\{-\mu\sqrt{\sigma2} + \mu\,\tau + \sqrt{\sigma2}\,M[1] - \tau\,M[1],\ \mu^2\sqrt{\sigma2} - \sigma2^{3/2} - \mu^2\,\tau + \sigma2\,\tau -$$
$$2\,\mu\sqrt{\sigma2}\,M[1] + 2\,\mu\,\tau\,M[1] + \sqrt{\sigma2}\,M[2] - \tau\,M[2],\ \sigma2 - 2\sqrt{\sigma2}\,\tau + \tau^2\Big\}\Big\}$$

$$\begin{pmatrix} \sigma2 & -\mu^3 - 3\,\mu\,\sigma2 + M[3] & -\frac{\mu^3+3\,\mu\,\sigma2-M[3]}{2\,\tau} \\ -\mu^3 - 3\,\mu\,\sigma2 + M[3] & 3\,\mu^4 + 6\,\mu^2\,\sigma2 - 2\,\sigma2\,\tau^2 + \tau^4 - 4\,\mu\,M[3] + M[4] & \frac{3\,\mu^4+6\,\mu^2\,\sigma2-2\,\sigma2\,\tau^2+\tau^4-4\,\mu\,M[3]+M[4]}{2\,\tau} \\ -\frac{\mu^3+3\,\mu\,\sigma2-M[3]}{2\,\tau} & \frac{3\,\mu^4+6\,\mu^2\,\sigma2-2\,\sigma2\,\tau^2+\tau^4-4\,\mu\,M[3]+M[4]}{2\,\tau} & \frac{3\,\mu^4+6\,\mu^2\,\sigma2-2\,\sigma2\,\tau^2+\tau^4-4\,\mu\,M[3]+M[4]}{4\,\tau^2} \end{pmatrix}$$

Direct use of a Δ-theorem

```
Clear[M]
g[a_, b_] = {a, b - a², √(b - a²)};
(J = (D[g[a, b], {{a, b}}] /. {a → M[1], b → M[2]})) // MatrixForm
Σ = Table[M[i + j] - M[i] M[j], {i, 1, 2}, {j, 1, 2}];
J.Σ.Transpose[J] // Simplify // MatrixForm
Ai.B.Transpose[Ai] /. {μ → M[1], σ2 → M[2] - M[1]², τ → √(M[2] - M[1]²)} // Simplify //
 MatrixForm
```

$$\begin{pmatrix} 1 & 0 \\ -2\,M[1] & 1 \\ -\frac{M[1]}{\sqrt{-M[1]^2+M[2]}} & \frac{1}{2\sqrt{-M[1]^2+M[2]}} \end{pmatrix}$$

$$\begin{pmatrix} -M[1]^2+M[2] & 2\,M[1]^3-3\,M[1]\,M[2]+M[3] & \frac{2\,M[1]^3-3}{2\sqrt{-I}} \\ 2\,M[1]^3-3\,M[1]\,M[2]+M[3] & -4\,M[1]^4+8\,M[1]^2\,M[2]-M[2]^2-4\,M[1]\,M[3]+M[4] & \frac{-4\,M[1]^4+8\,M[1]^2\,M[2}{2\sqrt{-I}} \\ \frac{2\,M[1]^3-3\,M[1]\,M[2]+M[3]}{2\sqrt{-M[1]^2+M[2]}} & \frac{-4\,M[1]^4+8\,M[1]^2\,M[2]-M[2]^2-4\,M[1]\,M[3]+M[4]}{2\sqrt{-M[1]^2+M[2]}} & \frac{4\,M[1]^4-8\,M[1]^2\,M[2}{4\,M[} \end{pmatrix}$$

$$\begin{pmatrix} -M[1]^2+M[2] & 2\,M[1]^3-3\,M[1]\,M[2]+M[3] & \frac{2\,M[1]^3-3}{2\sqrt{-I}} \\ 2\,M[1]^3-3\,M[1]\,M[2]+M[3] & -4\,M[1]^4+8\,M[1]^2\,M[2]-M[2]^2-4\,M[1]\,M[3]+M[4] & \frac{-4\,M[1]^4+8\,M[1]^2\,M[2}{2\sqrt{-I}} \\ \frac{2\,M[1]^3-3\,M[1]\,M[2]+M[3]}{2\sqrt{-M[1]^2+M[2]}} & \frac{-4\,M[1]^4+8\,M[1]^2\,M[2]-M[2]^2-4\,M[1]\,M[3]+M[4]}{2\sqrt{-M[1]^2+M[2]}} & \frac{4\,M[1]^4-8\,M[1]^2\,M[2}{4\,M[} \end{pmatrix}$$

Special case of normal distribution

```
M[k_] = Moment[NormalDistribution[μ, σ], k];
J.Σ.Transpose[J] // Simplify // MatrixForm
Ai.B.Transpose[Ai] /. {μ → M[1], σ2 → M[2] - M[1]², τ → √(M[2] - M[1]²)} // Simplify //
 MatrixForm
```

$$\begin{pmatrix} \sigma^2 & 0 & 0 \\ 0 & 2\,\sigma^4 & (\sigma^2)^{3/2} \\ 0 & (\sigma^2)^{3/2} & \frac{\sigma^2}{2} \end{pmatrix}$$

$$\begin{pmatrix} \sigma^2 & 0 & 0 \\ 0 & 2\,\sigma^4 & (\sigma^2)^{3/2} \\ 0 & (\sigma^2)^{3/2} & \frac{\sigma^2}{2} \end{pmatrix}$$

Example: Sample correlation coefficient as an M-estimator

For a general, properly integrable distribution. We find the ***joint distribution*** of the sample means of X and Y, sample variances of X and Y, sample average of XY, and the sample correlation coefficient.

For the joint distribution, WLOG centered distributions with unit variances does not apply (compare with the example Session1.nb). We must compute in full generality.

```
Clear[M];
$Assumptions = Element[μx, Reals] && Element[μy, Reals] &&
   Element[μxy, Reals] && σx2 > 0 && σy2 > 0 && -1 < ρ < 1;
ψ[μx_, μy_, μxy_, σx2_, σy2_, ρ_] = {x - μx, y - μy, x y - μxy,
   (x - μx)^2 - σx2, (y - μy)^2 - σy2, (μxy - μx μy)/√(σx2 σy2) - ρ};
(A0 = D[ψ[μx, μy, μxy, σx2, σy2, ρ], {{μx, μy, μxy, σx2, σy2, ρ}}]);
(A = A0 /. {x → M[1, 0], y → M[0, 1]});
(A = A /. μxy → M[1, 1]) // MatrixForm
(B0 = Outer[Times, ψ[μx, μy, μxy, σx2, σy2, ρ], ψ[μx, μy, μxy, σx2, σy2, ρ]]);
(B = (B0 // Expand) /. Flatten[Table[x^(4-j) y^(4-k) → M[4 - j, 4 - k], {j, 0, 4}, {k, 0, 4}]]);
(B = B /. μxy → M[1, 1] // Simplify) // MatrixForm
```

$$\begin{pmatrix} -1 & 0 & 0 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & -1 & 0 & 0 & 0 \\ -2\,(-\mu x + M[1,0]) & 0 & 0 & -1 & 0 & 0 \\ 0 & -2\,(-\mu y + M[0,1]) & 0 & 0 & -1 & 0 \\ -\frac{\mu y}{\sqrt{\sigma x2\,\sigma y2}} & -\frac{\mu x}{\sqrt{\sigma x2\,\sigma y2}} & \frac{1}{\sqrt{\sigma x2\,\sigma y2}} & -\frac{\sigma y2\,(-\mu x\,\mu y + M[1,1])}{2\,(\sigma x2\,\sigma y2)^{3/2}} & -\frac{\sigma x2\,(-\mu x\,\mu y + M[1,1])}{2\,(\sigma x2\,\sigma y2)^{3/2}} & -1 \end{pmatrix}$$

$$\begin{pmatrix} \mu x^2 - 2\,\mu x\,M[1,0] + M[2,0] & \cdots \\ \mu x\,(\mu y - M[0,1]) - \mu y\,M[1,0] + M[1,1] & \cdots \\ -M[1,0]\,M[1,1] + M[2,1] & \cdots \\ -\mu x^3 + 3\,\mu x^2\,M[1,0] - \sigma x2\,M[1,0] + \mu x\,(\sigma x2 - 3\,M[2,0]) + M[3,0] & -\sigma x2 \\ -\mu x\,(\mu y^2 - \sigma y2 - 2\,\mu y\,M[0,1] + M[0,2]) + \mu y^2\,M[1,0] - \sigma y2\,M[1,0] - 2\,\mu y\,M[1,1] + M[1,2] & \cdots \\ \frac{(\mu x - M[1,0])\,\left(\rho\,\sigma x2\,\sigma y2 + \sqrt{\sigma x2\,\sigma y2}\,(\mu x\,\mu y - M[1,1])\right)}{\sigma x2\,\sigma y2} & \cdots \end{pmatrix}$$

Asymptotic variance matrix for general distributions.

```
(Var = Inverse[A].B.Transpose[Inverse[A]] // Simplify) // MatrixForm
Var /. {μx → M[1, 0], μy → M[0, 1], σx2 → M[2, 0] - M[1, 0]^2, σy2 → M[0, 2] - M[0, 1]^2} //
  Simplify // MatrixForm
```

```
μx
2 √(σx2 σy2) (μx-M[1,0]) (ρ σx2 σy2+√(σx2 σy2) (μx μy-M[1,1]))-2 σx2 (μx σy2-(μy-M[0,1]) (μx μy-M[1,1])) (μx (μy-M[0,1])-μy M[
```

```
M[1,1] M[1,2] (M[1,0]^2-M[2,0])-M[0,1]^3 (M[1,0]^2 M[2,0]-2 M[2,0]^2+M[1,0] M[3,0])+M[0,2] (M[1,0]^3 M[1,1]-2 M[1,0]^2 M[2,1]+2 M[
```

Asymptotic variance of the sample correlation coefficient.

```
Clear[M]
Short[(Var[[6, 6]] /. {μx → M[1, 0], μy → M[0, 1],
      σx2 → M[2, 0] - M[1, 0]^2, σy2 → M[0, 2] - M[0, 1]^2}) // Simplify]
Var[[6, 6]] /. {μx → M[1, 0], μy → M[0, 1], σx2 → M[2, 0] - M[1, 0]^2,
   σy2 → M[0, 2] - M[0, 1]^2, ρ → 0} // Simplify
```

```
(M[0, 4] M[1, 1]^2 (M[1, 0]^2 - M[2, 0])^2 - 4 <<1>>^3 <<1>>^2 ((-1 + ρ^2) M[1, 0]^2 - ρ^2 M[2, 0]) +
   <<7>> + <<1>> - 2 M[0, 1] (<<1>>)) / (4 (M[0, 1]^2 - M[0, 2])^3 (M[1, 0]^2 - M[2, 0])^3)
```

```
                        1
─────────────────────────────────────────────
4 (M[0, 1]^2 - M[0, 2])^3 (M[1, 0]^2 - M[2, 0])^3
(M[0, 4] M[1, 1]^2 (M[1, 0]^2 - M[2, 0])^2 + 4 M[0, 2]^3 (M[1, 0]^3 - M[1, 0] M[2, 0])^2 +
  2 M[0, 2] M[1, 1] (M[1, 0]^2 - M[2, 0]) (2 M[1, 0] M[1, 1] M[1, 2] - 2 M[1, 0]^2 M[1, 3] +
     2 M[1, 3] M[2, 0] + 2 M[0, 3] (M[1, 0]^3 - M[1, 0] M[2, 0]) - M[1, 1] M[2, 2]) +
  M[0, 1]^6 (4 M[1, 0]^6 - 8 M[1, 0]^4 M[2, 0] + 4 M[2, 0]^3 - 4 M[1, 0] M[2, 0] M[3, 0] +
     M[1, 0]^2 (3 M[2, 0]^2 + M[4, 0])) +
  M[0, 2]^2 (-8 M[1, 0]^5 M[1, 2] + 4 M[1, 0]^3 (4 M[1, 2] M[2, 0] - 3 M[1, 1] M[2, 1]) +
     4 M[2, 0]^2 M[2, 2] + M[1, 0]^4 (11 M[1, 1]^2 + 4 M[2, 2]) - 4 M[1, 0]
      (2 M[1, 2] M[2, 0]^2 + M[1, 1] (-3 M[2, 0] M[2, 1] + M[1, 1] M[3, 0])) - 4 M[1, 1]
      M[2, 0] M[3, 1] - 4 M[1, 0]^2 (3 M[1, 1]^2 M[2, 0] + 2 M[2, 0] M[2, 2] - M[1, 1] M[3, 1]) +
     M[1, 1]^2 (4 M[2, 0]^2 + M[4, 0])) + M[0, 1]^4 (11 M[1, 1]^2 M[2, 0]^2 -
     4 M[1, 0]^3 (M[1, 2] M[2, 0] + M[1, 1] M[2, 1]) + 6 M[1, 0]^4 M[2, 2] + 4 M[2, 0]^2 M[2, 2] +
     4 M[1, 0] (M[1, 2] M[2, 0]^2 + M[1, 1] M[2, 0] M[2, 1] - M[1, 1]^2 M[3, 0]) - 4 M[1, 1]
      M[2, 0] M[3, 1] - 2 M[1, 0]^2 (4 M[1, 1]^2 M[2, 0] + 5 M[2, 0] M[2, 2] - 2 M[1, 1] M[3, 1]) +
     M[1, 1]^2 M[4, 0] - 2 M[0, 2] (4 M[1, 0]^6 - 7 M[1, 0]^4 M[2, 0] + 4 M[2, 0]^3 -
        4 M[1, 0] M[2, 0] M[3, 0] + M[1, 0]^2 (2 M[2, 0]^2 + M[4, 0]))) +
  M[0, 1]^2 ((M[1, 0]^2 - M[2, 0]) (M[0, 4] (M[1, 0]^4 - M[1, 0]^2 M[2, 0]) +
        2 M[1, 1] (-2 M[1, 0] M[1, 1] M[1, 2] + 2 M[1, 0]^2 M[1, 3] - 2 M[1, 3] M[2, 0] +
           2 M[0, 3] (M[1, 0]^3 - M[1, 0] M[2, 0]) + M[1, 1] M[2, 2])) + M[0, 2]^2 (3 M[1, 0]^6 -
        4 M[1, 0]^4 M[2, 0] + 4 M[2, 0]^3 - 4 M[1, 0] M[2, 0] M[3, 0] + M[1, 0]^2 M[4, 0]) +
     2 M[0, 2] (4 M[1, 0]^5 M[1, 2] + M[1, 0]^3 (-6 M[1, 2] M[2, 0] + 8 M[1, 1] M[2, 1]) -
        4 M[2, 0]^2 M[2, 2] - M[1, 0]^4 (4 M[1, 1]^2 + 5 M[2, 2]) +
        2 M[1, 0] (M[1, 2] M[2, 0]^2 + 2 M[1, 1] (-2 M[2, 0] M[2, 1] + M[1, 1] M[3, 0])) +
        4 M[1, 1] M[2, 0] M[3, 1] + M[1, 0]^2 (7 M[1, 1]^2 M[2, 0] + 9 M[2, 0] M[2, 2] -
           4 M[1, 1] M[3, 1]) - M[1, 1]^2 (6 M[2, 0]^2 + M[4, 0]))) -
  2 M[0, 1]^5 (4 M[1, 0]^5 M[1, 1] + 2 M[2, 0] (2 M[2, 0] M[2, 1] - M[1, 1] M[3, 0]) -
     2 M[1, 0]^2 (2 M[2, 0] M[2, 1] + M[1, 1] M[3, 0]) +
     2 M[1, 0]^3 (-5 M[1, 1] M[2, 0] + M[3, 1]) +
     M[1, 0] (-2 M[2, 0] M[3, 1] + M[1, 1] (9 M[2, 0]^2 + M[4, 0]))) +
  4 M[0, 1]^3 (-(M[1, 0]^2 - M[2, 0]) (M[1, 0]^2 M[1, 1] M[1, 2] + M[1, 0]^3 M[1, 3] +
        M[1, 1] (-3 M[1, 2] M[2, 0] + M[1, 1] M[2, 1]) +
        M[1, 0] (-2 M[1, 1]^3 - M[1, 3] M[2, 0] + M[1, 1] M[2, 2])) + M[0, 2]
      (5 M[1, 0]^5 M[1, 1] - M[1, 0]^4 M[2, 1] + 2 M[2, 0] (2 M[2, 0] M[2, 1] - M[1, 1] M[3, 0]) -
        M[1, 0]^2 (3 M[2, 0] M[2, 1] + 2 M[1, 1] M[3, 0]) + M[1, 0]^3 (-9 M[1, 1] M[2, 0] +
           2 M[3, 1]) + M[1, 0] (-2 M[2, 0] M[3, 1] + M[1, 1] (7 M[2, 0]^2 + M[4, 0])))) -
  2 M[0, 1] (M[1, 1] (M[0, 4] M[1, 0] + 2 M[0, 3] M[1, 1]) (M[1, 0]^2 - M[2, 0])^2 +
     2 M[0, 2] (M[1, 0]^2 - M[2, 0]) (-M[1, 0]^2 M[1, 1] M[1, 2] - M[1, 0]^3 M[1, 3] +
        M[0, 3] (M[1, 0]^4 - M[1, 0]^2 M[2, 0]) + M[1, 1] (3 M[1, 2] M[2, 0] - M[1, 1] M[2, 1]) +
        M[1, 0] (2 M[1, 1]^3 + M[1, 3] M[2, 0] - M[1, 1] M[2, 2])) +
     M[0, 2]^2 (9 M[1, 0]^5 M[1, 1] - 2 M[1, 0]^4 M[2, 1] +
        2 M[2, 0] (2 M[2, 0] M[2, 1] - M[1, 1] M[3, 0]) - 2 M[1, 0]^2
         (M[2, 0] M[2, 1] + M[1, 1] M[3, 0]) + 2 M[1, 0]^3 (-7 M[1, 1] M[2, 0] + M[3, 1]) +
        M[1, 0] (-2 M[2, 0] M[3, 1] + M[1, 1] (8 M[2, 0]^2 + M[4, 0])))))
```

Special case of the bivariate normal distribution.

```
dist = MultinormalDistribution[{μx, μy}, {{σx2, ρ √(σx2 σy2)}, {ρ √(σx2 σy2), σy2}}];
M[i_, j_] = Moment[dist, {i, j}];
Var // Simplify // MatrixForm
Var /. {μx → 0, μy → 0, σx2 → 1, σy2 → 1} // Simplify // MatrixForm
```

$$\begin{pmatrix} \sigma x2 & \rho\sqrt{\sigma x2\,\sigma y2} & \mu y\,\sigma x2 + \mu x\,\rho\sqrt{\sigma x2\,\sigma y2} \\ \rho\sqrt{\sigma x2\,\sigma y2} & \sigma y2 & \mu x\,\sigma y2 + \mu y\,\rho\sqrt{\sigma x2\,\sigma y2} \\ \mu y\,\sigma x2 + \mu x\,\rho\sqrt{\sigma x2\,\sigma y2} & \mu x\,\sigma y2 + \mu y\,\rho\sqrt{\sigma x2\,\sigma y2} & \mu y^2\,\sigma x2 + \left(\mu x^2 + \sigma x2 + \rho^2\,\sigma x2\right)\,\sigma y2 + 2\,\mu x\,\mu y\,\rho\,\sqrt{\sigma x} \\ 0 & 0 & 2\,\rho\,\sigma x2\sqrt{\sigma x2\,\sigma y2} \\ 0 & 0 & 2\,\rho\,\sigma y2\sqrt{\sigma x2\,\sigma y2} \\ 0 & 0 & -\left(-1+\rho^2\right)\sqrt{\sigma x2\,\sigma y2} \end{pmatrix}$$

$$\begin{pmatrix} 1 & \rho & 0 & 0 & 0 & 0 \\ \rho & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1+\rho^2 & 2\rho & 2\rho & 1-\rho^2 \\ 0 & 0 & 2\rho & 2 & 2\rho^2 & \rho-\rho^3 \\ 0 & 0 & 2\rho & 2\rho^2 & 2 & \rho-\rho^3 \\ 0 & 0 & 1-\rho^2 & \rho-\rho^3 & \rho-\rho^3 & \left(-1+\rho^2\right)^2 \end{pmatrix}$$

Direct use of the Δ-theorem (only for the asymptotic distribution of $\hat{\rho}$, see Session1.nb)

```
Clear[μ];
g[a_, b_, c_, d_, e_] = (e - a b) / ((c - a^2) (d - b^2))^(1/2);
Dg =
  D[g[a, b, c, d, e], {{a, b, c, d, e}}] /. {a → 0, b → 0, c → 1, d → 1, e → ρ} // Simplify;
MatrixForm[Dg] (* gradient of g at E(a,b,c,d,e) *)
Σ = {{1, ρ, μ[3, 0], μ[1, 2], μ[2, 1]}, {ρ, 1, μ[2, 1], μ[0, 3], μ[1, 2]},
   {μ[3, 0], μ[2, 1], μ[4, 0] - 1, μ[2, 2] - 1, μ[3, 1] - μ[1, 1]},
   {μ[1, 2], μ[0, 3], μ[2, 2] - 1, μ[0, 4] - 1, μ[1, 3] - μ[1, 1]},
   {μ[2, 1], μ[1, 2], μ[3, 1] - μ[1, 1], μ[1, 3] - μ[1, 1], μ[2, 2] - μ[1, 1]^2}};
(* original variance matrix *)
MatrixForm[Σ]
Dg . Σ . Dg // Simplify (* asymptotic variance of ρ̂ *)
Dg . Σ . Dg /. ρ → 0 (* asymptotic variance under independence *)
```

$$\begin{pmatrix} 0 \\ 0 \\ -\frac{\rho}{2} \\ -\frac{\rho}{2} \\ 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 & \rho & \mu[3,0] & \mu[1,2] & \mu[2,1] \\ \rho & 1 & \mu[2,1] & \mu[0,3] & \mu[1,2] \\ \mu[3,0] & \mu[2,1] & -1+\mu[4,0] & -1+\mu[2,2] & -\mu[1,1]+\mu[3,1] \\ \mu[1,2] & \mu[0,3] & -1+\mu[2,2] & -1+\mu[0,4] & -\mu[1,1]+\mu[1,3] \\ \mu[2,1] & \mu[1,2] & -\mu[1,1]+\mu[3,1] & -\mu[1,1]+\mu[1,3] & -\mu[1,1]^2+\mu[2,2] \end{pmatrix}$$

$$2\,\rho\,\mu[1,1] - \mu[1,1]^2 + \mu[2,2] - \rho\left(\mu[1,3] + \mu[3,1]\right) + \frac{1}{4}\rho^2\left(-4 + \mu[0,4] + 2\,\mu[2,2] + \mu[4,0]\right)$$

$$-\mu[1,1]^2 + \mu[2,2]$$

Example: Robust estimation of location

```
$Assumptions = Element[θ, Reals];
dist = {NormalDistribution[], StudentTDistribution[1],
   LaplaceDistribution[], LogisticDistribution[]};
ρ[x_, θ_] = Log[PDF[#, x - θ]] & /@ dist;
D[ρ[x, θ], θ]
ψ[x_, θ_] = %;
Plot[Evaluate[ψ[x, 3]], {x, 3 - 2, 3 + 2},
 PlotLegends → {"Normal", "Cauchy",  "Laplace", "Logistic"}, PlotLabel → "ψ functions"]
```

$$\left\{x-\theta,\ \frac{2\,(x-\theta)}{1+(x-\theta)^2},\ \frac{\begin{cases}\frac{e^{-x+\theta}}{2} & x-\theta\geq 0\\ -\frac{e^{x-\theta}}{2} & \text{True}\end{cases}}{\begin{cases}\frac{e^{-x+\theta}}{2} & x-\theta\geq 0\\ \frac{e^{x-\theta}}{2} & \text{True}\end{cases}},\ e^{x-\theta}\left(1+e^{-x+\theta}\right)^2\left(-\frac{2\,e^{-2x+2\theta}}{\left(1+e^{-x+\theta}\right)^3}+\frac{e^{-x+\theta}}{\left(1+e^{-x+\theta}\right)^2}\right)\right\}$$



Inverse asymptotic variance (Fisher information)

```
Clear[x]
-D[ψ[x, θ], {θ}] // Simplify
Table[Integrate[%[[j]] PDF[dist[[j]], x - θ], {x, -∞, ∞}], {j, 1, Length[dist]}]
```

$$\left\{1,\ -\frac{2\left(-1+x^2-2\,x\,\theta+\theta^2\right)}{\left(1+x^2-2\,x\,\theta+\theta^2\right)^2},\ 0,\ \frac{2\,e^{x+\theta}}{\left(e^x+e^\theta\right)^2}\right\}$$

$$\left\{1,\ \frac{1}{2},\ 0,\ \frac{1}{3}\right\}$$